



# INSTITUTO UNIVERSITARIO AERONÁUTICO

## Modelo Adiabático Ideal de Motores Stirling

**Informe Técnico:** DMA-003/12

**Revisión:** /

**Proyecto:** Stirling-Solar

**Fecha:** 28/02/2012

**Autor:**

Ing. Gustavo Scarpin  
*Director Dpto. Mec. Aer*



## **PROYECTO STIRLING-SOLAR**

### **Modelo Adiabático Ideal de Motores Stirling**

Por:

*Ing. Gustavo Scarpin*

#### **RESUMEN**

Dentro del proyecto de investigación sobre generación eléctrica mediante la combinación de un motor con ciclo Stirling y energía solar, se ha procedido a desarrollar un modelo adiabático ideal de motor Stirling.

Dicho modelo, descrito en detalle en este informe, permite analizar cualquiera de las geometrías de motores Stirling existentes.

Debido a las hipótesis realizadas es de esperar que el modelo sobreestime las performances reales del motor.

En el Anexo B se presenta el programa de cómputo StirAd.for el cual permite automatizar los cálculos. A modo de ejemplo se calcula las performances de un motor tipo Alfa.

**Córdoba, 28 de febrero de 2012**



**ÍNDICE**

<b>1. INTRODUCCIÓN</b> .....	<b>4</b>
<b>2. DESARROLLO</b> .....	<b>5</b>
2.1 Principales Hipótesis .....	5
2.2 Desarrollo del grupo de ecuaciones .....	5
2.2.1 Conservación de la masa .....	5
2.2.2 Conservación de energía .....	6
2.2.3 Transferencia de calor en el modelo de motor adiabático .....	7
2.2.4 Obtención de las ecuaciones diferenciales relevantes .....	8
2.2.4.1 Zona de compresión .....	8
2.2.4.2 Zona de expansión.....	9
2.2.4.3 Regenerador .....	9
2.2.4.4 Zona de calentamiento.....	10
2.2.4.5 Zona de enfriamiento .....	10
2.2.4.6 Ecuación de variación de presión.....	11
2.2.4.7 Trabajo.....	11
2.3 Resumen de ecuaciones .....	12
2.3.1 Modelo .....	12
2.3.2 Presión.....	12
2.3.3 Masas .....	12
2.3.4 Acumulación de masa .....	13
2.3.5 Caudal másico .....	13
2.3.6 Condicional de temperatura .....	13
2.3.7 Energía .....	13
2.4 Consideraciones varias .....	14
<b>3. RESULTADOS</b> .....	<b>16</b>
3.1 Diagrama Presión-Volumen.....	16
3.2 Presión en función del ángulo de manivela .....	17
3.3 Volumen total en función del ángulo de manivela.....	18
3.4 Temperaturas en las zonas de compresión y expansión .....	18
3.5 Flujos de Calor en las Zona de Enfriamiento, Regeneración y Calentamiento .....	19
3.6 Caudales Másicos entre Zonas .....	20



# INSTITUTO UNIVERSITARIO AERONÁUTICO

## INFORME TÉCNICO DMA-003/12

Página 3 de 34

---

3.7	Performances del motor .....	21
4.	<b>CONCLUSIONES</b> .....	<b>22</b>
5.	<b>REFERENCIAS</b> .....	<b>22</b>
	<b>ANEXO A: TEMPERATURA MEDIA EFECTIVA DEL REGENERADOR</b> .....	<b>23</b>
	<b>ANEXO B: MANUAL DE USUARIO – PROGRAMA STIRAD</b> .....	<b>25</b>

1. **INTRODUCCIÓN**

Existe una gran variedad de modelos matemáticos para estimar el comportamiento físico de los motores Stirling, desde los más simples, como el modelo isotérmico hasta los que modelan prácticamente todo el motor mediante CFD.

El Modelo Adiabático Ideal que aquí se desarrolla es uno de los más simples, pero a diferencia del modelo isotérmico permite estimar en forma preliminar la transferencia de calor en los distintos componentes.

Por lo general a todas las configuraciones de motores Stirling se los puede analizar dividiéndolos en 5 zonas conectadas en forma lineal tal como se muestra en la Figura 1.




Figura 1: Motores Stirling - Configuraciones más comunes<sup>1</sup>.

El enfriador, regenerador, y calentador tiene volúmenes constante  $V_k$ ,  $V_r$ , y  $V_h$  respectivamente. Tanto la zona de compresión como la de expansión son cada una divididas en un volumen de holgura fija,  $V_{ck}$  y  $V_{cle}$  respectivamente, y un volumen que depende del tiempo variando entre 0 y sus respectivos valores máximos,  $V_{swc}$  y  $V_{swe}$ . La variación exacta de  $V_{swc}$  y  $V_{swe}$  durante un ciclo completo depende de las transmisiones mecánicas empleadas.

El Tipo-Alfa puede ser modelado colocando las cinco zonas, una detrás de otra, en un solo cilindro. En el Tipo-Beta tanto el pistón de potencia como el desplazador se alojan en un solo cilindro, mientras que en la Tipo-Gamma ocupan cilindros separados.



## 2. DESARROLLO

### 2.1 Principales Hipótesis

- a. El motor gira a velocidad constante, por lo tanto, el ángulo de rotación de la manivela  $\theta$  es proporcional al tiempo.
- b. Todas las pérdidas de presión debido a la fricción y a las diferencias de presión necesarias para acelerar el flujo del gas de trabajo son despreciadas. Por tal motivo, la presión  $p$ , tiene el mismo valor en todo el motor y varía sólo con el tiempo. Consecuentemente, en la ecuación de conservación de energía, es despreciada la energía cinética del gas de trabajo.
- c. Se asumen despreciables las fugas de gas hacia el exterior.
- d. Se asume que las zonas de compresión y expansión, con volúmenes  $V_c$  y  $V_e$ , son adiabáticas. Esto significa que no existe intercambio de calor entre el gas y su entorno, tanto sea las superficies de las paredes como en las de las caras del pistón/desplazador. Las temperaturas en cada uno de estos espacios es uniforme pero varían durante el ciclo debido a los cambios de presión, volumen y masa del gas procedente de la zona del enfriador y calentador respectivamente.
- e. Las condiciones de transferencia de calor en la zona del enfriador es lo suficientemente buena como para mantener el gas dentro del volumen  $V_k$  a la temperatura  $T_c$  uniforme y constante en todo momento. Lo mismo es cierto para la zona del calentador de volumen  $V_h$ , siendo la temperatura del mismo  $T_h$  uniforme y constante.
- f. Las condiciones de transferencia de calor son suficientes para mantener lineal la distribución de temperaturas dentro del regenerador, el volumen  $V_r$ , variando de  $T_c$ , donde el regenerador está conectado al enfriador, hasta  $T_h$  en el lado del calentador.
- g. Se asume que el gas de trabajo es un gas ideal, es decir:

$$pV = mRT$$

$$c_p - c_v = R, \quad \gamma = c_p / c_v$$

$$u = c_v T, \quad h = c_p T$$

### 2.2 Desarrollo del grupo de ecuaciones

#### 2.2.1 Conservación de la masa

El balance de masa para cualquier sistema sometido a cualquier proceso puede ser expresado como<sup>2</sup>:

$$m_{in} - m_{out} = \Delta m_{system}$$

Donde  $\Delta m_{system} = m_{final} - m_{initial}$  es el cambio en la masa del sistema. Para volúmenes de control (CV), también puede ser expresado más explícitamente como:

$$m_i - m_e = (m_2 - m_1)_{CV}$$

Donde:  $i$  = entrada,  $e$  = salida,  $1$  = estado inicial del CV,  $2$  = estado final del CV.

### 2.2.2 Conservación de energía




Figura 2: Motores Stirling – Modelo general para las 5 zonas<sup>3</sup>

De la primera ley de la termodinámica:

$$\underbrace{E_{in} - E_{out}}_{\text{Energía neta transferida por calor, trabajo y masa.}} = \underbrace{\Delta E_{system}}_{\text{Cambio en las energías interna, cinética y potencial}}$$

El proceso de flujo inestacionario general, por lo general, es dificultoso de analizar debido a que las propiedades de la masa en la entrada y salida pueden cambiar durante un proceso. Sin embargo, la mayoría de los procesos de flujo inestacionario, pueden ser representados razonablemente bien mediante un proceso de flujo uniforme, el cual involucra la siguiente idealización:

*Las propiedades del fluido, a lo largo de la sección de una entrada o salida, son uniformes y constantes, es decir, no cambian con el tiempo o la posición. Si lo hacen, se promedian y se tratan como constantes durante todo el proceso.*

El balance de energía para un sistema de flujo uniforme puede ser expresado explícitamente como<sup>2</sup>:

$$\left[ Q_{in} + W_{in} + \sum_{in} m(h + ke + pe) \right] - \left[ Q_{out} + W_{out} + \sum_{out} m(h + ke + pe) \right] = \left[ m_2(u + ke + pe)_2 - m_1(u + ke + pe)_1 \right]_{system}$$

Cuando los cambios de energía cinética y potencial asociado con el volumen de control y flujos de fluidos son insignificantes, como suele ser generalmente, el balance de energía se simplifica a:

$$\Delta Q - \Delta W = \sum_{out} m h - \sum_{in} m h + (m_2 u_2 - m_1 u_1)_{system}$$

Con

$$\Delta Q = Q_{net,in} = Q_{in} - Q_{out}$$

$$\Delta W = W_{net,in} = W_{in} - W_{out}$$

### 2.2.3 Transferencia de calor en el modelo de motor adiabático

La ecuación general de energía puede escribirse como:

$$DQ + (c_p T_{in} m_{in} - c_p T_{out} m_{out}) = dW + c_v D(mT)$$

Asumiendo gas ideal:

$$pV = mRT$$

Aplicando el logaritmo y diferenciando:

$$\frac{Dp}{p} + \frac{DV}{V} = \frac{Dm}{m} + \frac{DT}{T}$$




Figura 3: Motores Stirling – Modelo Adiabático Ideal<sup>3</sup>

La masa total de trabajo permanece constante, se manera que:

$$m_c + m_k + m_r + m_h + m_e = M$$

Diferenciando la ec. de masa total:

$$Dm_c + Dm_k + Dm_r + Dm_h + Dm_e = 0$$

De la ecuación de masa total y de la ec. de estado se tiene:

$$p \frac{\left( \frac{V_c}{T_c} + \frac{V_k}{T_k} + \frac{V_r}{T_r} + \frac{V_h}{T_h} + \frac{V_e}{T_e} \right)}{R} = M$$

Según el modelo de temperatura presentado en Figura 3, la temperatura media efectiva del regenerador resulta (ver Anexo A):

$$T_r = \frac{T_h - T_k}{\ln \left( \frac{T_h}{T_k} \right)}$$

Despejando la presión de la ecuación anterior,

$$p = \frac{M R}{\frac{V_c}{T_c} + \frac{V_k}{T_k} + \frac{V_r}{T_r} + \frac{V_h}{T_h} + \frac{V_e}{T_e}}$$

## 2.2.4 Obtención de las ecuaciones diferenciales relevantes

### 2.2.4.1 Zona de compresión




Figura 4: Modelo Zona de Compresión<sup>3</sup>

De la ecuación de estado:  $m_c = \frac{p V_c}{R T_c}$

Por otro lado, aplicando la ecuación de energía a esta zona, se tiene:

$$-c_p T_{ck} \dot{m}_{ck} = dW_c + c_v D(m_c T_c)$$

Considerando la ecuación de continuidad, la velocidad de acumulación de gas es igual a la masa entrante, es decir:  $Dm_c = -\dot{m}_{ck}$ , y el trabajo realizado es dado por  $dW_c = p DV_c$ .

$$c_p T_{ck} Dm_c = p DV_c + c_v D(m_c T_c)$$

De la ecuación anterior y con las tres primeras relaciones para gas ideal, listadas anteriormente, operando:

$$Dm_c = \frac{\gamma p DV_c + V_c Dp}{\gamma R T_{ck}}$$



Para tener en cuenta la diferencia de temperatura entre la zona del compresor y el calentador, se tiene:

$$\text{If } (\dot{m}_{ck} > 0) \text{ then } T_{ck} = T_c \text{ else } T_{ck} = T_k$$

De la forma diferencial de la ecuación de estado:

$$DT_c = T_c \left( \frac{Dp}{p} + \frac{DV_c}{V_c} - \frac{Dm_c}{m_c} \right)$$

#### 2.2.4.2 Zona de expansión

Operando de igual forma que en la zona de compresión se tiene:

$$m_e = \frac{p V_e}{R T_e}$$

$$Dm_e = \dot{m}_{he}$$

$$Dm_e = \frac{\gamma p DV_e + V_e Dp}{\gamma R T_{he}}$$

Con

$$\text{If } (\dot{m}_{he} > 0) \text{ then } T_{he} = T_h \text{ else } T_{he} = T_e$$

$$DT_e = T_e \left( \frac{Dp}{p} + \frac{DV_e}{V_e} - \frac{Dm_e}{m_e} \right)$$

#### 2.2.4.3 Regenerador

De la ecuación de estado:  $m_r = \frac{p V_r}{R T_r}$

La temperatura efectiva del regenerador es (Anexo A):

$$T_r = \frac{T_h - T_k}{\ln \left( \frac{T_h}{T_k} \right)}$$

Este valor estacionario de temperatura puede ser usado para calcular la masa dentro del regenerador, asumiendo gas ideal y teniendo en cuenta la distribución lineal de temperatura. Con el volumen  $V_r$  permaneciendo constante, un cambio en la masa es relacionado simplemente al cambio en la presión:

$$dm_r = \frac{m_r}{p} dp = \frac{V_r}{R T_r} dp$$



El flujo másico resulta (caudal entrante +):

$$\dot{m}_{kr} = \dot{m}_{ck} - dm_k$$

$$\dot{m}_{rh} = \dot{m}_{he} + dm_h$$

De la ecuación de energía, siendo el trabajo nulo:

$$DQ_r + c_p (T_{kr} \dot{m}_{kr} - T_{rh} \dot{m}_{rh}) = c_v T_r Dm = c_v \frac{V_r}{R} Dp$$

Reordenando

$$DQ_r = c_v \frac{V_r}{R} Dp - c_p (T_{kr} \dot{m}_{kr} - T_{rh} \dot{m}_{rh})$$

Otra aproximación es la presentada en Ref. [1]:

$$DQ_r = \frac{[V_r + \gamma(V_c + V_k + V_h + V_e)] Dp + \gamma p (DV_e + DV_c)}{\gamma - 1}$$

#### 2.2.4.4 Zona de calentamiento

Las ecuaciones son similares a las deducidas para el regenerador, sólo que en este caso la temperatura es impuesta. Las ecuaciones para esta zona son:

$$m_h = \frac{p V_h}{R T_h}$$

$$dm_h = \frac{m_h}{p} dp = \frac{V_h}{R T_h} dp$$

$$DQ_h = c_v \frac{V_h}{R} Dp - c_p (T_{rh} \dot{m}_{rh} - T_{he} \dot{m}_{he})$$

#### 2.2.4.5 Zona de enfriamiento

Las condiciones, y por lo tanto las ecuaciones resultantes, son similares a las anteriores:

$$m_k = \frac{p V_k}{R T_k}$$

$$dm_k = \frac{m_k}{p} dp = \frac{V_k}{R T_k} dp$$

$$DQ_k = c_v \frac{V_k}{R} Dp - c_p (T_{ck} \dot{m}_{ck} - T_{kr} \dot{m}_{kr})$$



### 2.2.4.6 Ecuación de variación de presión

De la ec.

$$Dm_c + Dm_k + Dm_r + Dm_h + Dm_e = 0$$

Reemplazando los diferenciales de masa:

$$\frac{\gamma p DV_c + V_c Dp}{\gamma R T_{ck}} + \frac{V_k}{R T_k} Dp + \frac{V_r}{R T_r} Dp + \frac{V_h}{R T_h} Dp + \frac{\gamma p DV_e + V_e Dp}{\gamma R T_{he}} = 0$$

Despejando el diferencial de presión se obtiene:

$$Dp = \frac{-\gamma p \left( \frac{DV_c}{T_{ck}} + \frac{DV_e}{T_{he}} \right)}{\left[ \frac{V_c}{T_{ck}} + \gamma \left( \frac{V_k}{T_k} + \frac{V_r}{T_r} + \frac{V_h}{T_h} \right) + \frac{V_e}{T_{he}} \right]}$$

### 2.2.4.7 Trabajo

Finalmente el trabajo realizado en las celdas de compresión y expansión está dado por:

$$W = W_c + W_e$$

$$DW_c = p DV_c$$

$$DW_e = p DV_e$$

$$DW = DW_c + DW_e$$

Reemplazando:

$$DW = p DV_c + p DV_e$$

## 2.3 Resumen de ecuaciones

### 2.3.1 Modelo




Figura 5: Motores Stirling – Modelo Adiabático Ideal<sup>3</sup>

### 2.3.2 Presión

$$p = \frac{M R}{\frac{V_c}{T_c} + \frac{V_k}{T_k} + \frac{V_r}{T_r} + \frac{V_h}{T_h} + \frac{V_e}{T_e}}$$

$$Dp = \frac{-\gamma p \left( \frac{DV_c}{T_{ck}} + \frac{DV_e}{T_{he}} \right)}{\left[ \frac{V_c}{T_{ck}} + \gamma \left( \frac{V_k}{T_k} + \frac{V_r}{T_r} + \frac{V_h}{T_h} \right) + \frac{V_e}{T_{he}} \right]}$$

### 2.3.3 Masas

$$m_c = \frac{p V_c}{R T_c}$$

$$m_k = \frac{p V_k}{R T_k}$$

$$m_r = \frac{p V_r}{R T_r}$$

$$m_h = \frac{p V_h}{R T_h}$$

$$m_e = \frac{p V_e}{R T_e}$$



### 2.3.4 Acumulación de masa

$$Dm_c = \frac{\gamma p DV_c + V_c Dp}{\gamma R T_{ck}}$$

$$Dm_k = \frac{V_k}{R T_k} Dp$$

$$Dm_r = \frac{V_r}{R T_r} Dp$$

$$Dm_h = \frac{V_h}{R T_h} Dp$$

$$Dm_e = \frac{\gamma p DV_e + V_e Dp}{\gamma R T_{he}}$$

### 2.3.5 Caudal másico

$$Dm_c = -\dot{m}_{ck}$$

$$\dot{m}_{kr} = \dot{m}_{ck} - dm_k$$

$$\dot{m}_{rh} = \dot{m}_{he} + dm_h$$

$$Dm_e = \dot{m}_{he}$$

### 2.3.6 Condicional de temperatura

$$\text{If } (\dot{m}_{ck} > 0) \text{ then } T_{ck} = T_c \text{ else } T_{ck} = T_k$$

$$\text{If } (\dot{m}_{he} > 0) \text{ then } T_{he} = T_h \text{ else } T_{he} = T_e$$

### 2.3.7 Energía

$$DW_c = p DV_c$$

$$DQ_k = c_v \frac{V_k}{R} Dp - c_p (T_{ck} \dot{m}_{ck} - T_{kr} \dot{m}_{kr})$$

$$DQ_r = c_v \frac{V_r}{R} Dp - c_p (T_{kr} \dot{m}_{kr} - T_{rh} \dot{m}_{rh})$$

$$DQ_h = c_v \frac{V_h}{R} Dp - c_p (T_{rh} \dot{m}_{rh} - T_{he} \dot{m}_{he})$$

$$DW_e = p DV_e$$

$$DW = DW_c + DW_e$$

$$W = W_c + W_e$$



El sistema está formado por 22 variables y 16 derivadas a ser resueltas en un ciclo completo ( $\theta = [0, 2\pi]$ ):

- 7 derivadas a ser integradas numéricamente:  $T_c, T_e, Q_k, Q_r, Q_h, W_c, W_e$
- 9 variables y derivadas analíticas:  $W, p, V_c, V_e, m_c, m_k, m_r, m_h, m_e$
- 6 variables condicionales y de flujo másico (derivadas indefinidas):  
 $T_{ck}, T_{he}, \dot{m}_{ck}, \dot{m}_{kr}, \dot{m}_{rh}, \dot{m}_{he}$

## 2.4 Consideraciones varias

1. El modelo representa un sistema de flujo cuasi-estacionario, en el cual, entre los intervalos de integración, se asume que las cuatro variables de flujo másico ( $\dot{m}_{ck}, \dot{m}_{kr}, \dot{m}_{rh}, \dot{m}_{he}$ ) permaneces constante, es decir, no existen efectos por la aceleración.
2. Si los intercambiadores de calor son isotérmicos y el regenerador es ideal, se tiene por definición que:

$$T_{kr} = T_k$$
$$T_{rh} = T_h$$

3. El modelo Adiabático Ideal no es un modelo de valor inicial sino que es un problema de condiciones de contorno. Por lo tanto no se conocen los valores iniciales de  $T_c$  y  $T_e$ , los cuales resultan de los procesos de compresión y expansión adiabáticos, así como de los procesos de flujos entálpicos. Para obtenerlos se asignan valores arbitrarios como condiciones iniciales y se espera que los valores de los mismos sean iguales al principio ( $\theta = 0^\circ$ ) y final del ciclo ( $\theta = 360^\circ$ ) (Por ejemplo con un error relativo de  $10^{-5}$ ). Generalmente se suele asignar como condición inicial:

$$T_{c_0} = T_k$$
$$T_{e_0} = T_h$$

Generalmente es necesario entre 5 y 10 ciclos para que el sistema converja.

4. Se suele usar Runge-Kutta de 4to orden para integrar el sistema hasta lograr las condiciones estacionarias (repetitividad entre dos ciclos). Se lo suele verificar calculando la presión promedio del ciclo y verificando que sea igual a la presión promedio propuesta en la entrada (por ej. 20 Bar).



5. En el caso de movimiento sinusoidal:

$$V_c = V_{clc} + \frac{V_{swc}}{2}(1 + \cos(\theta))$$

$$V_e = V_{cle} + \frac{V_{swe}}{2}(1 + \cos(\theta + \delta))$$

$$dV_c = -\frac{V_{swc}}{2} \sin(\theta) d\theta$$

$$dV_e = -\frac{V_{swe}}{2} \sin(\theta + \delta) d\theta$$

Donde  $\delta$  es el desfase de volumen (VLP).

6. Los datos de entrada son:

- i. Gas de trabajo: Define  $\gamma$  y  $R$ , ó,  $c_p$  y  $c_v$
- ii. Presión promedio,  $P_{ave}$
- iii. Temperatura del calentador,  $T_h$
- iv. Temperatura del enfriador,  $T_k$
- v. Desfase de volumen,  $\delta$
- vi. Volumen libre de compresión,  $V_{clc}$
- vii. Volumen de barrido de compresión,  $V_{swc}$
- viii. Volumen del enfriador,  $V_k$
- ix. Volumen del regenerador,  $V_r$
- x. Volumen del calentador,  $V_h$
- xi. Volumen de barrido de expansión,  $V_{swe}$
- xii. Volumen libre de expansión,  $V_{cle}$
- xiii. Velocidad de rotación de la manivela,  $\dot{\theta}$



### 3. RESULTADOS

El procedimiento descrito en §2 fue implementado en un programa de cómputo, el cual se lista en el ANEXO B.

A modo de ejemplo de utilización y con el propósito de validar los resultados se ha modelado el motor Ross D-90<sup>4</sup> con movimiento sinusoidal.

Tabla 1: Motor Ross D-90<sup>4</sup>

Gas de trabajo		Air	
Presión promedio	$P_{ave}$	200000	Pa
Temperatura del calentador	$T_h$	923	K
Temperatura del enfriador	$T_k$	300	K
Desfasaje de volumen	$\delta$	95.5694	°
Volumen libre de compresión	$V_{clc}$	8.0000	cm <sup>3</sup>
Volumen de barrido de compresión	$V_{swc}$	61.0450	cm <sup>3</sup>
Volumen del enfriador	$V_k$	31.2101	cm <sup>3</sup>
Volumen del regenerador	$V_r$	34.8885	cm <sup>3</sup>
Volumen del calentador	$V_h$	28.5093	cm <sup>3</sup>
Volumen de barrido de expansión	$V_{swe}$	61.0450	cm <sup>3</sup>
Volumen libre de expansión	$V_{cle}$	10.0000	cm <sup>3</sup>
Velocidad de rotación de la manivela	$\dot{\theta}$	40.0000	1/sec

Seguidamente se presentan las performances de dicho motor obtenidas con el modelo adiabático ideal descrito en este informe.

Los resultados coinciden con los presentados en Ref. [3] excepto para las performances del regenerador, ya que en dicha referencia se utiliza un método alternativo de cálculo.

#### 3.1 Diagrama Presión-Volumen

La curva que se muestra en la Figura 6 es típica para casi todos los motores de combustión interna o externa. El área encerrada por el bucle es igual al trabajo neto por ciclo producido por el motor (3.706480 J), mientras que el área entre el eje  $V_{tot}$  y la parte superior del bucle se corresponde con el calor que se ha consumido por ciclo. La relación entre los dos es igual a la eficiencia real del motor, aquí 0.625116. Este valor es, por supuesto, menor que la eficiencia de Carnot (0.674973) basado en las temperaturas del enfriador y calentador. La razón termodinámica de esta disminución en la eficiencia, es la mezcla del gas que sale de la zona de compresión y expansión, respectivamente, a temperaturas diferentes de la del gas en la zona de enfriamiento y calentamiento, respectivamente.




Figura 6: Diagrama Presión-Volumen

### 3.2 Presión en función del ángulo de manivela

Como era de esperarse la presión fluctúa entre un mínimo (1.433 bar en  $\theta = 340^\circ$ ) y un máximo (2.733 bar en  $\theta = 160^\circ$ ). La ubicación precisa es, por supuesto, determinada por la cinemática empleada (en este caso sinusoidal) y los tamaños de las zonas de enfriamiento, regeneración y calentamiento.




Figura 7: Presión en función del ángulo de manivela

### 3.3 Volumen total en función del ángulo de manivela

Esto simplemente refleja el tipo de cinemática empleada y el tamaño de las zonas de enfriamiento, regeneración y calentamiento. Como se puede esperar, la relación entre el volumen máximo y mínimo afecta fuertemente la relación de compresión, entre la presión máxima y la mínima.




Figura 8: Volumen en función del ángulo de manivela

### 3.4 Temperaturas en las zonas de compresión y expansión




Figura 9: Temperaturas en las zonas de compresión y expansión



Debido a la falta de transferencia de calor en estas zonas las temperaturas del gas oscilan entre un mínimo y máximo durante un ciclo completo, aunque los gases que entran en estos espacios desde las zonas de enfriamiento y calentamiento, entrar a temperaturas fijas ( $T_k$  y  $T_h$ , respectivamente). La razón de las oscilaciones en las temperaturas son los cambios de presión.

### 3.5 Flujos de Calor en las Zona de Enfriamiento, Regeneración y Calentamiento

Las curvas mostradas representan el calor que fluye en estas zonas (positivo cuando el calor fluye desde las paredes hacia el gas, de lo contrario negativo) durante el tiempo que ha transcurrido entre que la manivela rota de la posición  $\theta = 0^\circ$  y  $\theta$ , la referida en el eje horizontal.

Para la zona de enfriamiento esto significa que unos 6.08 [J] de calor se elimina del gas durante los primeros 170° (los valores de los ángulos son sólo aproximados) de rotación del cigüeñal (hasta que se alcanza el mínimo en  $Q_k$ ). Durante el resto de 190° para completar un ciclo completo unos 3.86 [J] se añaden de nuevo al gas resultando una pérdida neta de calor de 2.22 [J]. Por lo tanto, en cierta medida el enfriador actúa como un recuperador, adicionando calor al gas durante una parte de ciclo y la eliminando calor en otras. Es importante notar que la cantidad total de calor intercambiado en la zona de enfriamiento (6.08 + 3.86 julios) es en un factor 4.5 veces mayor que el calor neto evacuado de 2.22 [J] lo cual deberá tenerse en cuenta al diseñar el enfriador.

En la zona de calentamiento, el comportamiento es similar pero no tan dramático. Durante los primeros 40° de la rotación del cigüeñal se añade calor (0.86 [J]), durante los próximos 130° se elimina calor (2,60 [J]), mientras que entre  $\theta = 170^\circ$  y  $360^\circ$  se añaden unos 7.67 [J] de calor resultando una transferencia de calor neta hacia el gas de 5.93 [J]. Por lo tanto, la cantidad total de calor transferido,  $11.3 = 0.86 + 2.60 + 7.67$  [J], es sólo alrededor de dos veces mayor que el calor neto incorporado.




Figura 10: Flujos de Calor



Para el regenerador se ve que se elimina calor del gas durante los primeros  $55^\circ$  (hasta alcanzar el mínimo). Luego se transfiere calor al gas desde los  $55^\circ$  a los  $210^\circ$ , el cual es eliminado entre  $210^\circ$  y  $360^\circ$  con una transferencia de calor neto de cero. El calor añadido durante una parte del ciclo y que luego se retira durante la parte restante es de  $32.47 \text{ [J]}$  lo cual es aproximadamente 9 (nueve) veces mayor que el trabajo neto producido por el motor de  $3.7 \text{ [J]}$ .

### 3.6 Caudales Másicos entre Zonas

Las tasas de flujo de masas por unidad son proporcionales a la velocidad de rotación del motor. Como tal, las curvas en la Figura 11 se escalan linealmente con la velocidad de rotación, pero conservan su forma y tamaño con respecto a la otra. El flujo másico desde la zona de compresión hacia el enfriador se asume positivo cuando el gas fluye desde la zona de compresión a la de enfriamiento y negativo cuando el gas fluye en sentido contrario. La misma convención se utiliza para los otros flujos másicos.

Al observar el gráfico, lo más obvio son los dos puntos en el que las 4 curvas se cruzan, es decir, cuando los 4 tipos de flujo de masa son iguales en valor. En  $\theta = 160^\circ$  el gas fluye por todas partes desde la zona de compresión en la dirección a la zona de expansión con la presión alcanzando su máximo. En  $\theta = 340^\circ$  el caudal es nuevamente igual en todas partes, pero en dirección opuesta con la presión en su mínimo. Para el modelo adiabático ideal de los motores Stirling, se puede demostrar que estos dos puntos existen, y que siempre coinciden con la ubicación de la presión mínima y máxima.

En el gráfico, también se puede notar que al regenerador en un momento dado le ingresa gas por ambos extremos (calentador y enfriador) (desde los  $40^\circ$  donde la línea  $kr$  corta el eje de abscisa y los  $60^\circ$  donde la línea  $rh$  corta dicho eje). En forma similar existe un intervalo (entre  $200^\circ$  y  $220^\circ$ ) en donde del regenerador sale gas de ambos extremos. Análisis similares se puede hacer para el enfriador y calentador.




Figura 11: Flujo Másico entre Zonas



### 3.7 Performances del motor

Seguidamente se muestra en la Tabla 2 los datos de performance que pueden ser obtenidos del análisis, para los datos de entrada propuestos en Tabla 1.

**Tabla 2: Performances Motor Stirling**

<b>Características del Motor</b>	
Presión [Bar]	: mín 1.434 , máx 2.733 , Promedio 2.000
Trabajo/ciclo [J]	: 3.706
Potencia Neta [W]	: 148.25
Eficiencia de Carnot [-]	: 0.675
Eficiencia Real [-]	: 0.625
Masa de gas [g]	: 0.2495
<b>Zona de Compresión</b>	
Temperatura [K]	: mín. 279.5, máx 337.6
Trabajo [J]	: -2.223
Calor [J]	: 0.000
<b>Zona del Enfriador</b>	
Trabajo [J]	: 0.000
Calor [J]	: -2.222
Flujo de Calor [W]	: mín. -923.5, máx 555.6
<b>Zona del regenerador</b>	
Temp. Efectiva [K]	: 554.3
Trabajo [J]	: 0.000
Calor Aportado al Gas [J]	: 32.5
Calor Extraído al Gas [J]	: -32.5
Calor Neto [J]	: 0.000
Flujo de Calor [W]	: mín. -3280, máx. 5104



**Zona del Calentador**

Trabajo [J]	:	0.000
Calor [J]	:	5.925
Flujo de Calor [W]	:	mín. -496.5, máx. 760.8

**Zona de Expansión**

Temperatura [K]	:	mín. 779.7, máx. 939.4
Trabajo [J]	:	5.929
Calor [J]	:	0.000

**4. CONCLUSIONES**

En el marco del proyecto Stirling-Solar se ha desarrollado un programa de cómputos que permite predecir el comportamiento global y de cada una de las partes de un motor con ciclo Stirling.

El programa resuelve las ecuaciones de un modelo adiabático ideal de motor Stirling.

Este tipo de programa es una herramienta importante ya que permite modificar las principales variables de diseño y rápidamente ver cómo ellas afectan al rendimiento del motor.

Debido a las hipótesis realizadas para simplificar el modelo es de esperar que las performances calculadas sean sobreestimadas.

**5. REFERENCIAS**

- [1] Siegfried Herzog, "Ideal Adiabatic Simulation of Stirling Engines", Feb-2012.  
<http://mac6.ma.psu.edu/stirling/simulations/IdealAdiabatic/index.html>
- [2] Yunus A. Cengel, Michael A. Boles, "Thermodynamics: An Engineering Approach", McGraw-Hill; Fifth edition, ISBN-10: 0070606595, 2006.
- [3] Israel Urieli, "Stirling Cycle Machine Analysis - Ideal Adiabatic Analysis", Feb-2012.  
<http://www.ohio.edu/mechanical/stirling/adiabatic/adiabatic.html>
- [4] Andy Ross, "Making Stirling Engines", Ross Experimental, 3<sup>er</sup> Edition, Jun-1997.

## ANEXO A: Temperatura Media Efectiva del Regenerador

Con el fin de evaluar correctamente la masa de gas en el volumen del regenerador, se debe conocer la distribución de la temperatura a lo largo del mismo. En este sentido, se supondrá que el regenerador ideal tiene un perfil de temperatura lineal entre la temperatura fría  $T_k$  y la temperatura  $T_h$  caliente, tal como se muestra en la Figura A-1.




Figura A-1: Perfil Lineal de Temperatura del Regenerador

En la figura anterior se puede observar que el perfil de temperatura a lo largo del regenerador puede ser expresada como:

$$T(x) = \frac{(T_h - T_k)}{L_r} x + T_k$$

donde  $L_r$  es la longitud del regenerador.

La masa total de gas,  $m_r$ , dentro del volumen  $V_r$  del regenerador, está dada por:

$$m_r = \int_0^{V_r} \rho dV_r$$

donde  $\rho$  es la densidad.

Asumiendo un regenerador de área constante  $A_r$ , se tiene:

$$dV_r = A_r dx$$

y

$$V_r = A_r L_r$$



Sustituyendo  $V_r$  y  $dV_r$ , y asumiendo válida la ecuación de estado,  $P = \rho R T$ , la integral anterior se puede escribir de la siguiente forma:

$$m_r = \frac{V_r P}{R} \int_0^{L_r} \frac{1}{(T_h - T_k) x + T_k} dx$$

Integrando y simplificando:

$$m_r = \frac{V_r P}{R} \frac{\ln\left(\frac{T_h}{T_k}\right)}{(T_h - T_k)}$$

Por otro lado, como se ha definido en el texto principal, la temperatura media efectiva del gas en el regenerador,  $T_r$ , en términos de la ecuación de estado para un gas ideal es

$$m_r = \frac{V_r P}{R T_r}$$

Comparando las dos ecuaciones anteriores se obtiene finalmente la temperatura media efectiva del regenerador,  $T_r$ , como función de  $T_k$  y  $T_h$ .

$$T_r = \frac{(T_h - T_k)}{\ln\left(\frac{T_h}{T_k}\right)}$$



## ANEXO B: Manual de Usuario – Programa StirAd

### B-1 INTRODUCCION

El programa StirAd es un programa en Fortran que permite calcular las performances de un motor Stirling asumiendo un Modelo Adiabático Ideal. El modelo matemático fue desarrollado en la primera parte de este informe. En este Anexo se presenta el listado fuente del programa junto con un archivo de entrada de ejemplo y su correspondiente archivo de salida.

### B-2 ARCHIVO DE ENTRADA

El archivo de entrada StirAd.dat consta de un solo Namelist, DATOS, con la siguiente información:

- i. P, Presión promedio [Pa]
- ii. T<sub>k</sub>, Temperatura del enfriador [K]
- iii. T<sub>h</sub>, Temperatura del calentador [K]
- iv. V<sub>clc</sub>, Volumen libre de compresión [m<sup>3</sup>]
- v. V<sub>cle</sub>, Volumen libre de expansión [m<sup>3</sup>]
- vi. V<sub>swc</sub>, Volumen de barrido de compresión [m<sup>3</sup>]
- vii. V<sub>swe</sub>, Volumen de barrido de expansión [m<sup>3</sup>]
- viii. V<sub>k</sub>, Volumen del enfriador [m<sup>3</sup>]
- ix. V<sub>r</sub>, Volumen del regenerador [m<sup>3</sup>]
- x. V<sub>h</sub>, Volumen del calentador [m<sup>3</sup>]
- xi. Omega, Velocidad de rotación de la manivela [Hz]
- xii. Delta, Desfasaje de volumen [°]
- xiii. cambio\_theta, Paso de ángulo de manivela [°]
- xiv. Error\_max, Error máximo admisible en la presión media [Pa]

```
&DATOS
P      = 2.D5
T_k    = 300.D0
T_h    = 923.D0
V_clc  = 8.0000D-6
V_cle  = 10.0000D-6
V_swc  = 61.0450D-6
V_swe  = 61.0450D-6
V_k    = 31.2101D-6
V_r    = 34.8885D-6
V_h    = 28.5093D-6
omega  = 40.D0
delta  = 95.5694D0
cambio_theta = 5.D0
Error_max = 1.D-10
/
```



### B-3 ARCHIVO DE SALIDA

En el archivo de salida StirAd.out se imprime el Namelist con los datos de entrada y seguidamente se imprime una tabla conteniendo la siguiente información ordenada por columnas:

- i. Theta [deg], Angulo de manivela (de 0° a 360° con variación de paso cambio\_theta)
- ii. p[bar], Presión en el motor
- iii. Tc[K], Temperatura en la zona de compresión
- iv. Te[K], Temperatura en la zona de expansión
- v. Vc[cc], Volumen de la zona de compresión
- vi. Ve[cc], Volumen de la zona de expansión
- vii. Vtot[cc], Volumen total (Vc+Vk+Vr+Vh+Ve)
- viii. Wc[J], Trabajo del pistón de la zona de compresión (Pistón de desplazamiento)
- ix. We[J], Trabajo del pistón de la zona de expansión (Pistón de trabajo)
- x. Wtot[J], Trabajo total (la suma de ambos)
- xi. Qk[J], Calor entrante (+) o saliente (-) de la zona de enfriamiento
- xii. Qr[J], Calor entrante (+) o saliente (-) de la zona de regeneración
- xiii. Qh[J], Calor entrante (+) o saliente (-) de la zona de calentamiento
- xiv. dWc[W], Variación de trabajo en el tiempo en la zona de compresión
- xv. dWe[W], Variación de trabajo en el tiempo en la zona de expansión
- xvi. dWtot[W], Variación de trabajo en el tiempo total (la suma de ambos)
- xvii. dQk[W], Variación de calor en el tiempo en la zona de enfriamiento
- xviii. dQr[W], Variación de calor en el tiempo en la zona de regeneración
- xix. dQh[W], Variación de calor en el tiempo en la zona de calentamiento
- xx. mc[g], Masa en la zona de compresión
- xxi. mk[g], Masa en la zona de enfriamiento
- xxii. mr[g], Masa en la zona de regeneración
- xxiii. mh[g], Masa en la zona de calentamiento
- xxiv. me[g], Masa en la zona de expansión
- xxv. mtot[g], Masa total en el sistema (suma de las anteriores = cte)
- xxvi. ck[g/s], Caudal o flujo másico desde la zona de compresión a la de enfriamiento
- xxvii. kr[g/s], Caudal o flujo másico desde la zona de enfriamiento a la de regeneración
- xxviii. rh[g/s], Caudal o flujo másico desde la zona de regeneración a la de calentamiento
- xxix. he[g/s], Caudal o flujo másico desde la zona de calentamiento a la de expansión

Por cuestiones de espacio la tabla resultado se muestra en forma comprimida y luego de ser levantada a una planilla Excel.



Table with 17 columns: Theta[gr], P[Bar], Tc[K], Te[K], Vc[cc], Ve[cc], Vtot[cc], Wc[J], We[J], Wtot[J], Qk[J], Qr[J], Qh[J], dWc[W], dWe[W], dWtot[W]. Rows include values for Theta from 0 to 360.

Table with 14 columns: Theta[gr], dQk[W], dQr[W], dQh[W], mc[g], mk[g], mr[g], mh[g], me[g], mtot[g], ck[g/s], kr[g/s], rh[g/s], he[g/s]. Rows include values for Theta from 0 to 360.

B-4 LISTADO FUENTE

Seguidamente se lista el programa fuente y sus subrutinas. Las variables son definidas mediante un include.



**PROGRAMA PRINCIPAL: StirAd**

```
*****
**                               Modelo Adiabático de un motor Stirling                               **
*****
** Tema       : Calculo de las performances de un motor Stirling                               **
** Descripcion: Modelo adiabático                                                                 **
*****
** Programa Principal : StirAd.for                                                                **
** Autor              : Gustavo Scarpin                                                            **
** Fecha              : 27 de Diciembre de 2011                                                  **
** Última modificación: 27 de Diciembre de 2011                                                  **
*****
```

```
PROGRAM StirAd
IMPLICIT NONE
INCLUDE 'StirAd.inc'

REAL*8 T,W_tot,V_tot,PRESi,PRESf
REAL*8 W_c0, W_e0, Q_k0, Q_r0, Q_h0, theta0, cambio_W

OPEN(UNIT=3,FILE='StirAd.out',STATUS='UNKNOWN')

C  ENTRADA DE VALORES
CALL lee_entrada

WRITE(3,1000)

PRESi = P

DO WHILE (DABS(PRESf-PRESi).GT.Error_max)
PRESf = 0.D0
theta0 = -cambio_theta
DO theta=0.D0,360.D0*10.D0,cambio_theta
T = theta/(360.d0*omega)

CALL simula_estado(T)

C  IMPRIME SALIDA
IF (theta.ge.3240.D0) then
IF (theta.eq.3240.D0) then
W_c0 = W_c
W_e0 = W_e
Q_k0 = Q_k
Q_r0 = Q_r
Q_h0 = Q_h
ENDIF

PRESf = PRESf + P

V_tot = V_c+V_k+V_r+V_h+V_e
W_tot = W_c+W_e
cambio_W = cambio_W_c+cambio_W_e

theta0=theta0+cambio_theta

WRITE(3, '(30G15.5)')theta0,P/1.D5,T_c,T_e,V_c/1.D-6,V_e/1.D-6,
. V_tot/1.D-6,W_c-W_c0,W_e-W_e0,
. W_tot-(W_c0+W_e0),Q_k-Q_k0,Q_r-Q_r0,Q_h-Q_h0,
. cambio_W_c,cambio_W_e,cambio_W,cambio_Q_k,cambio_Q_r,cambio_Q_h,
. m_c*1.D3,m_k*1.D3,m_r*1.D3,m_h*1.D3,m_e*1.D3,masa*1.D3,
. m_punto_ck*1.D3,m_punto_kr*1.D3,m_punto_rh*1.D3,m_punto_he*1.D3

WRITE(*, '(15G15.5)')theta0,P
ENDIF
END DO

PRESf = (PRESf-P)/(360.D0/cambio_theta) !Cálculo de la Presión Promedio

Masa = Masa * PRESi/PRESf
WRITE(*,*) PRESi,PRESf,masa
END DO

1000 FORMAT(T1,' Theta[gr] P[Bar] Tc[K] Te[K]
```



```
.]      Vc[cc]      Ve[cc]      Vtot[cc]      Wc[J]
.      We[J]      Wtot[J]      Qk[J]      Qr[J]      Q
.h[J]      dWc[W]      dWe[W]      dWtot[W]      dQk[W]
.      dQr[W]      dQh[W]      mc[g]      mk[g]
.      mr[g]      mh[g]      me[g]      mtot[g]      ck[g/s]
.]      kr[g/s]      rh[g/s]      he[g/s]')
```

END PROGRAM StirAd

### SUBROUTINAS

```
*****
**      Modelo Adiabático de un motor Stirling      **
*****
** Tema      : Lee y escribe en archivos externos      **
** Descripción: Pre y post procesadores e inicio de variables      **
*****
** Subrutina      : Entradas_Salidas.for      **
** Programa Principal : StirAd.for      **
** Autor      : Gustavo Scarpin      **
** Fecha      : 27 de Diciembre de 2011      **
** Última modificación: 27 de Diciembre de 2011      **
*****
```

```
*****
**      ENTRADA DE DATOS      **
*****
```

```
SUBROUTINE lee_entrada
  IMPLICIT NONE
  INCLUDE 'StirAd.inc'

  NAMELIST/DATOS/T_k,T_h,V_clc,V_cle,V_swc,V_swe,V_k,V_r,V_h,P,
.      omega,delta,cambio_theta,Error_max

  OPEN(UNIT=1,FILE='StirAd.dat', STATUS='OLD')

  READ(1,DATOS)
  WRITE(3,DATOS)

C      TEMPERATURA MEDIA EN EL REGENERADOR
  T_r = (T_h-T_k)/LOG(T_h/T_k)

C      TEMPARATURA INICIAL EN c Y e
  T_c = T_k
  T_e = T_h
  T_ck = T_c
  T_he = T_h

C      INICIA VALORES DEL GAS
  CALL GAS

C      INICIA VALORES GEOMETRICOS
  DT = cambio_theta/(360.D0*omega)

  theta = 0.D0
  CALL GEOMETRIA

C      CALCULO DE LA MASA
  Masa = P/R*(V_c/T_c+V_k/T_k+V_r/T_r+V_h/T_h+V_e/T_e)

  CLOSE(1)

  RETURN
END SUBROUTINE lee_entrada
```

```
*****
**      GEOMETRIA DE LAS ZONAS c Y e      **
*****
```



```
SUBROUTINE geometria
  IMPLICIT NONE
  INCLUDE 'StirAd.inc'

  V_c = V_clc+V_sw/2.D0*(1+DCOSD(theta))
  V_e = V_cle+V_swe/2.D0*(1+DCOSD(theta+delta))

  cambio_V_c = -V_sw/2.D0*DSIND(theta)*omega*(2.D0*PI)
  cambio_V_e = -V_swe/2.D0*DSIND(theta+delta)*omega*(2.D0*PI)

  RETURN
END SUBROUTINE geometria
```

```
*****
**  DEFINE TIPO DE GAS                                     **
*****
```

```
SUBROUTINE gas
  IMPLICIT NONE
  INCLUDE 'StirAd.inc'

  Tipo_Gas = 'Aire'
  R = 287.15 D0
  gamma = 1.4D0

  c_v = R/(gamma - 1.D0)
  c_p = gamma * c_v

  RETURN
END SUBROUTINE gas
```

```
*****
**                      Modelo Adiabático de un motor Stirling          **
*****
** Tema      : Ecuaciones de estado del Stirling                       **
** Descripción: Modelo adiabático                                       **
*****
** Subrutina   : Ecuaciones_Estado.for                                  **
** Programa Principal : StirAd.for                                     **
** Autor      : Gustavo Scarpin                                       **
** Fecha     : 27 de Diciembre de 2011                               **
** Última modificación: 27 de Diciembre de 2011                       **
*****
```

```
SUBROUTINE simula_estado(T)
  IMPLICIT NONE
  INCLUDE 'StirAd.inc'

  REAL*8 T,Xe(7),XDe(7)
```

```
*****
**  ENTRADA ECUACIONES DE ESTADO                                     **
*****
```

```
Xe(1) = T_c
Xe(2) = T_e
Xe(3) = Q_k
Xe(4) = Q_r
Xe(5) = Q_h
Xe(6) = W_c
Xe(7) = W_e
```

```
CALL RK4(T,DT,Xe,XDe)
```

```
T_c      = Xe(1)
T_e      = Xe(2)
Q_k      = Xe(3)
Q_r      = Xe(4)
Q_h      = Xe(5)
W_c      = Xe(6)
W_e      = Xe(7)
```



```
cambio_T_c = XDe(1)
cambio_T_e = XDe(2)
cambio_Q_k = XDe(3)
cambio_Q_r = XDe(4)
cambio_Q_h = XDe(5)
cambio_W_c = XDe(6)
cambio_W_e = XDe(7)

RETURN
END      SUBROUTINE simula_estado

*****
**      ENTRADA OBTENCION DERIVADAS DE LAS ECUACIONES DE ESTADO      **
*****

SUBROUTINE derivadas_estado(T,Xe,XDe)
IMPLICIT NONE
INCLUDE 'StirAd.inc'

REAL*8 T,Xe(7),XDe(7)

C  ASIGNAR LAS VARIABLES DE ESTADO
T_c = Xe(1); T_e = Xe(2);
Q_k = Xe(3); Q_r = Xe(4); Q_h = Xe(5)
W_c = Xe(6); W_e = Xe(7)
T = T

CALL actualiza_variables

C  ECUACIONES DE CAMBIO DE TEMPERATURA
XDe(1) = T_c*(cambio_P/P + cambio_V_c/V_c - cambio_m_c/m_c)
XDe(2) = T_e*(cambio_P/P + cambio_V_e/V_e - cambio_m_e/m_e)

C  ECUACIONES DE INTERCAMBIO DE CALOR
XDe(3) = c_v*V_k/R*cambio_P-c_p*(T_ck*m_punto_ck-T_kr*m_punto_kr)
XDe(4) = c_v*V_r/R*cambio_P-c_p*(T_kr*m_punto_kr-T_rh*m_punto_rh)
XDe(5) = c_v*V_h/R*cambio_P-c_p*(T_rh*m_punto_rh-T_he*m_punto_he)

C  ECUACIONES DE CAMBIO DE TRABAJO
XDe(6) = P*cambio_V_c
XDe(7) = P*cambio_V_e

RETURN
END SUBROUTINE derivadas_estado

*****
**      ACTUALIZA VARIABLES      **
*****

SUBROUTINE actualiza_variables
IMPLICIT NONE
INCLUDE 'StirAd.inc'

C  VARIABLES GEOMETRICAS
CALL GEOMETRIA

C  VARIABLES AUXILIARES
P = Masa*R/(V_c/T_c+V_k/T_k+V_r/T_r+V_h/T_h+V_e/T_e)

C  Estimación previa
cambio_P = -gamma*P*(cambio_V_c/T_ck+cambio_V_e/T_he)/
.          (V_c/T_ck+gamma*(V_k/T_k+V_r/T_r+V_h/T_h)+V_e/T_he)
cambio_m_c = (gamma*P*cambio_V_c+V_c*cambio_P)/(gamma*R*T_ck)
cambio_m_e = (gamma*P*cambio_V_e+V_e*cambio_P)/(gamma*R*T_he)
m_punto_ck = -cambio_m_c
m_punto_he = cambio_m_e

C  Asignación de temperaturas
IF (m_punto_ck.GT.0.D0) THEN
T_ck = T_c
ELSE
T_ck = T_k
ENDIF
```



```
IF (m_punto_he.GT.0.D0) THEN
  T_he = T_h
ELSE
  T_he = T_e
ENDIF

T_kr = T_k
T_rh = T_h

C  VARIABLES PRINCIPALES

C  Cambio de la presión
cambio_P = -gamma*P*(cambio_V_c/T_ck+cambio_V_e/T_he)/
.          (V_c/T_ck+gamma*(V_k/T_k+V_r/T_r+V_h/T_h)+V_e/T_he)

C  Acumulación de masa
cambio_m_c = (gamma*P*cambio_V_c+V_c*cambio_P)/(gamma*R*T_ck)
cambio_m_k = V_k/(R*T_k)*cambio_P
cambio_m_r = V_r/(R*T_r)*cambio_P
cambio_m_h = V_h/(R*T_h)*cambio_P
cambio_m_e = (gamma*P*cambio_V_e+V_e*cambio_P)/(gamma*R*T_he)

C  Masas
m_c = P*V_c/(R*T_c)
m_k = P*V_k/(R*T_k)
m_r = P*V_r/(R*T_r)
m_h = P*V_h/(R*T_h)
m_e = P*V_e/(R*T_e)

C  Caudal másico
m_punto_ck = -cambio_m_c
m_punto_he = cambio_m_e
m_punto_kr = m_punto_ck - cambio_m_k
m_punto_rh = m_punto_he + cambio_m_h

RETURN
END SUBROUTINE actualiza_variables

SUBROUTINE RK4(TT,DT,XX,XD)
IMPLICIT NONE
INTEGER M,NX
PARAMETER (NX=7)
REAL*8 XX(NX),XD(NX),X(NX),XA(NX),T,TT,DT,Q

CALL derivadas_estado(TT,XX,XD)

DO M=1,NX
  XA(M) = XD(M)*DT
  X(M) = XX(M)+0.5D0*XA(M)
ENDDO
T = TT+0.5D0*DT
CALL derivadas_estado(T,X,XD)

DO M=1,NX
  Q = XD(M)*DT
  X(M) = XX(M)+0.5D0*Q
  XA(M) = XA(M)+Q+Q
ENDDO
CALL derivadas_estado(T,X,XD)

DO M=1,NX
  Q = XD(M)*DT
  X(M) = XX(M)+Q
  XA(M) = XA(M)+Q+Q
ENDDO
TT = TT+DT
CALL derivadas_estado(TT,X,XD)

DO M=1,NX
  XX(M) = XX(M)+(XA(M)+XD(M)*DT)/6.D0
ENDDO
```



RETURN

END SUBROUTINE RK4

**INCLUDE**

```
*****
**                               Modelo Adiabático de un motor Stirling          **
*****
** Tema       : Definicion de variables                                     **
** Descripcion: Definicion de variables globales                             **
*****
** Incluir    : StirAd.inc                                                 **
** Programa Principal : StirAd.for                                         **
** Autor      : Gustavo Scarpin                                           **
** Fecha      : 27 de Diciembre de 2011                                    **
** Ultima modificación: 27 de Diciembre de 2011                            **
*****
```

C DEFINICION DE CONSTANTES GLOBALES

```
REAL*8, PARAMETER :: PI = 3.14159265359D0
REAL*8, PARAMETER :: g = 9.80665D0
REAL*8, PARAMETER :: grad_rad = 0.01745329252D0
REAL*8, PARAMETER :: rad_grad = 57.2957795131D0
```

C VARIABLES DE ESTADO

```
REAL*8 T_c
REAL*8 T_e
REAL*8 Q_k
REAL*8 Q_r
REAL*8 Q_h
REAL*8 W_c
REAL*8 W_e
REAL*8 cambio_T_c
REAL*8 cambio_T_e
REAL*8 cambio_Q_k
REAL*8 cambio_Q_r
REAL*8 cambio_Q_h
REAL*8 cambio_W_c
REAL*8 cambio_W_e
```

```
COMMON/estado/T_c,T_e,Q_k,Q_r,Q_h,W_c,W_e,cambio_T_c,
.cambio_T_e,cambio_Q_k,cambio_Q_r,cambio_Q_h,cambio_W_c,cambio_W_e
```

C VARIABLES SUBROUTINA geometria

```
REAL*8 V_c
REAL*8 V_e
REAL*8 V_clc
REAL*8 V_cle
REAL*8 V_swc
REAL*8 V_swe
REAL*8 cambio_V_c
REAL*8 cambio_V_e
REAL*8 theta
REAL*8 cambio_theta
REAL*8 delta
```

```
COMMON/geo/V_c,V_e,V_clc,V_cle,V_swc,V_swe,cambio_V_c,cambio_V_e,
.theta,cambio_theta,delta
```

C VARIABLES SUBROUTINA gas

```
CHARACTER*10 Tipo_Gas
REAL*8 R
REAL*8 gamma
REAL*8 c_p
REAL*8 c_v
```

```
COMMON/var_gas/Tipo_Gas,R,gamma,c_p,c_v
```

C VARIABLES SUBROUTINA lee\_entrada

```
REAL*8 T_k
REAL*8 T_h
```



```
REAL*8 T_r
REAL*8 V_k
REAL*8 V_r
REAL*8 V_h
REAL*8 P
REAL*8 omega
REAL*8 DT
REAL*8 Masa
REAL*8 Error_max
```

```
COMMON/entrada/T_k,T_h,T_r,V_k,V_r,V_h,P,omega,DT,Masa,Error_max
```

```
C VARIABLES SUBROUTINA actualiza_variables
```

```
REAL*8 cambio_P
```

```
REAL*8 m_c
```

```
REAL*8 m_k
```

```
REAL*8 m_r
```

```
REAL*8 m_h
```

```
REAL*8 m_e
```

```
REAL*8 cambio_m_c
```

```
REAL*8 cambio_m_k
```

```
REAL*8 cambio_m_r
```

```
REAL*8 cambio_m_h
```

```
REAL*8 cambio_m_e
```

```
REAL*8 m_punto_ck
```

```
REAL*8 m_punto_he
```

```
REAL*8 m_punto_kr
```

```
REAL*8 m_punto_rh
```

```
REAL*8 T_ck
```

```
REAL*8 T_he
```

```
REAL*8 T_kr
```

```
REAL*8 T_rh
```

```
COMMON/variables/cambio_P,m_c,m_k,m_r,m_h,m_e,cambio_m_c,
.cambio_m_k,cambio_m_r,cambio_m_h,cambio_m_e,m_punto_ck,m_punto_he,
.m_punto_kr,m_punto_rh,T_ck,T_he,T_kr,T_rh
```